
====================EDGE_DETECT_TESTS====================================

CONVERGENCE OFTEN REQUIRES TRIGGERING OFF OF MULTI-EDGES.
HAVING BREAK BEFORE MAKE PULSES CAN PREVENT RACE CONDITIONS.

EDGE_DETECT_TESTS
*=========Create_Signal==================
.OPTIONS  GMIN=1p       METHOD=trap   ABSTOL=1m     TEMP=27   srcsteps = 1  gminsteps = 1 
.OPTIONS  RELTOL=.001   ABSTOL=1p     VNTOL=1u      ITL4=500  ITL1=400 
*V_PULSE# NODE_P NODE_N DC     VALUE  PULSE( VINIT  VPULSE TDELAY TRISE  TFALL  PWIDTH PERIOD )
VSQ       SQ     0      DC     0      PULSE( 0      1      1u     1u     1u     1m     2m )

XPOS_E    SQ     PEJ     POS_E 
XNEG_E    SQ     NEJ     NEG_E
XBOTH_E   SQ     BEJ     BOTH_E
XBRKMAK   SQ     BRK     MAK      BRKMAK

.control
set       pensize = 2
tran      1u    10m      0       1u
plot      sq    pej-.1   nej-.2  bej-.3
plot      brk   mak-.1
.endc

*=========Break_Make============================================
*
*                                                      |\     ___
*  ___   |\             VBF ____               DPE   __|3\___|BRK|
* |IN |__|3\________________|   \ PE                |  | /   |___|
* |___|  | / |           VLP| PE \___/\  /\  /\_____|  |/
*        |/  |/\  /\  /\____|    /     \/  \/   |   |
*               \/  \/   |  |___/      RLP2    _|_  |  |\     ___
*               RLP     _|_                    ___  |__|3\___|MAK|
*                       ___                CLP2 |      | /   |___|
*                   CLP  |                     _|_     |/
*                       _|_                    ///
*                       ///

*
.SUBCKT   BRKMAK  IN     BRK    MAK   
 BBUF     VBF     0      V =    u( v(IN )-.5 )
 RLP      VBF     VLP    10k
 CLP      VLP     0      50n  
 BAND     VPE     0      V =    u( u(v(VBF )-.5)*u(.5 -v(VLP ) ) -.5)
 RLP2     VPE     VLP2   3k
 CLP2     VLP2    0      20n 
 BBRK     BRK     0      V =    1-u(v(VLP2 )-.2)
 BMAK     MAK     0      V =    u(v(VLP2 )-.9)
.ENDS     BRKMAK

*=========BOTH_Edge============================================
*
*  ___   |\             VBF ____                ___
* |IN |__|3\________________\   \   ___    ___  |  \   ___
* |___|  | / |           VLP |PE \_|OUT|  |IN |_|PE \_|OUT|
*        |/  |/\  /\  /\_____|   / |___|  |___| |   / |___|
*               \/  \/   |   /__/               |__/
*               RLP     _|_
*                       ___
*                   CLP  |
*                       _|_
*                       ///
*
.SUBCKT   BOTH_E  IN     OUT   
 BBUF     VBF     0      V =   u( v(IN )-.5 )
 RLP      VBF     VLP    10k
 CLP      VLP     0      50n  
 BNOR     VNE     0      V =   1-u( u(v(VBF )-.5)+u(.5 -v(VLP ) ) -.5)
 BAND     VPE     0      V =   u( u(v(VBF )-.5)*u(.5 -v(VLP ) ) -.5)
 BOR      OUT     0      V =   u(u(v(VNE )-.5) + u(v(VPE )-.5)-.5)
.ENDS     BOTH_E



*=========NEG_Edge============================================
*
*  ___   |\             VBF ____                ___
* |IN |__|3\________________\   \   ___    ___  |  \   ___
* |___|  | / |           VLP |PE \_|OUT|  |IN |_|PE \_|OUT|
*        |/  |/\  /\  /\_____|   / |___|  |___| |   / |___|
*               \/  \/   |   /__/               |__/
*               RLP     _|_
*                       ___
*                   CLP  |
*                       _|_
*                       ///
*
.SUBCKT   NEG_E  IN     OUT   
 BBUF     VBF    0      V =   u( v(IN )-.5 )
 RLP      VBF    VLP    10k
 CLP      VLP    0      50n
 BNOR     OUT    0      V =   1-u( u(v(VBF )-.5)+u(.5 -v(VLP ) ) -.1)
.ENDS     NEG_E

*=========POS_Edge============================================
*
*  ___   |\             VBF ____                ___
* |IN |__|3\________________|   \   ___    ___  |  \   ___
* |___|  | / |           VLP| PE \_|OUT|  |IN |_|PE \_|OUT|
*        |/  |/\  /\  /\____|    / |___|  |___| |   / |___|
*               \/  \/   |  |___/               |__/
*               RLP     _|_
*                       ___
*                   CLP  |
*                       _|_
*                       ///
*
.SUBCKT   POS_E  IN     OUT   
 BBUF     VBF    0      V =   u( v(IN )-.5 )
 RLP      VBF    VLP    10k
 CLP      VLP    0      50n   IC=0
 BAND     OUT    0      V =   u( u(v(VBF )-.5)*u(.5 -v(VLP ) ) -.1)
.ENDS     POS_E

.end


